

Animated Graphs using SAS/Graph

VASUG – October 14, 2009

Jim Alessio

State Council of Higher Education for Virginia

Overview



"'Cheshire Puss,' [Alice] began... 'Would you tell me, please, which way I ought to go from here?'

'That depends a good deal on where you want to get to,' said the Cat.

'I don't much care where--' said Alice.

'Then it doesn't matter which way you go,' said the Cat.


'--so long as I get SOMEWHERE,' Alice added as an explanation.

'Oh, you're sure to do that,' said the Cat, 'if you only walk long enough.'"

SAS Devices


Device	Features	Requirements
ACTIVEX	Pop-up data tips, drill-down links, interactivity via pop-up menus	Must be viewed on a Windows system with the SAS ActiveX control installed locally
JAVA	Pop-up data tips, drill-down links, interactivity via pop-up menus	The Java 1.4 plug-in must be installed
JAVAMETA	Pop-up data tips, drill-down links, some interactivity such as zooming, and slide shows	SAS Metaview applet required
GIFANIM	Slide show of static images, no interactivity	None

Animated GIFs



Grade Distribution

Assessment= Test 1



SAS Code

```
filename plot1 'file location & name.gif';

goptions reset=all device=gifanim gsfname=plot1 gsfmode=replace delay=100
         cback=white cpattern=white ctext=black ftext=swissb ftitle=swissb
         xpixels=800 ypixels=600 iteration=0 gepilog='3B'X transparency
         disposal=background;

axis1 order=(0 to 25 by 5) minor=none label=(angle=90 'Number');

pattern c=blue v=solid;

proc gchart data=grades;
  by category notsorted;
  vbar grade / discrete freq=n axis=axis1;
  title1 'Grade Distribution';
  label grade='Grade'
        category='Assessment';
run;

quit;
```

GOPTIONS Components

```
goptions reset=all device=gifanim gsfname=plot1 gsfmode=replace delay=100  
cback=white cpattern=white ctext=black ftext=swissb ftitle=swissb  
xpixels=800 ypixels=600 iteration=0 gepilog='3B'X transparency  
disposal=background;
```

DEVICE = SAS/Graph device driver. The GIFANIM device driver enables you to create sequences of images that are displayed automatically from a single GIF file. These animated sequences are commonly referred to as slide shows.

GOPTIONS Components

```
goptions reset=all device=gifanim gsfname=plot1 gsfmode=replace delay=100  
        cback=white cpattern=white ctext=black ftext=swissb ftitle=swissb  
        xpixels=800 ypixels=600 iteration=0 gepilog='3B'X transparency  
        disposal=background;
```

GSFNAME - Graphic Stream File reference for output

GSFMODE - Replace existing file

GOPTIONS Components

```
goptions reset=all device=gifanim gsfname=plot1 gsfmode=replace delay=100  
cback=white cpattern=white ctext=black ftext=swissb ftitle=swissb  
xpixels=800 ypixels=600 iteration=0 gepilog='3B'X transparency  
disposal=background;
```

DELAY - specifies the length of time between graphs in units of 0.01 seconds. 100 = 1 second delay

GOPTIONS Components

```
goptions reset=all device=gifanim gsfname=plot1 gsfmode=replace delay=100  
cback=white cpattern=white ctext=black ftext=swissb ftitle=swissb  
xpixels=800 ypixels=600 iteration=0 gepilog='3B'X transparency  
disposal=background;
```

CBACK = background color

CPATTERN = default pattern color

CTEXT = default text color

FTEXT = default text font

FTITLE = default first title font

GOPTIONS Components

```
goptions reset=all device=gifanim gsfname=plot1 gsfmode=replace delay=100  
cback=white cpattern=white ctext=black ftext=swissb ftitle=swissb  
xpixels=800 ypixels=600 iteration=0 gepilog='3B'X transparency  
disposal=background;
```

XPIXELS = width in pixels

YPIXELS = height in pixels

Maximum for GIFANIM = 1280 x 1024

Larger values create larger output files

GOPTIONS Components

```
goptions reset=all device=gifanim gsfname=plot1 gsfmode=replace delay=100  
        cback=white cpattern=white ctext=black ftext=swissb ftitle=swissb  
        xpixels=800 ypixels=600 iteration=0 gepilog='3B'X transparency  
        disposal=background;
```

ITERATION = number of times to repeat the animation loop. A value of 0 causes the animation to loop continuously.

GOPTIONS Components

```
goptions reset=all device=gifanim gsfname=plot1 gsfmode=replace delay=100  
cback=white cpattern=white ctext=black ftext=swissb ftitle=swissb  
xpixels=800 ypixels=600 iteration=0 gepilog='3B'X transparency  
disposal=background;
```

GEPILOG = Sends a string to a device or file after all graphics commands are sent. '3B'X required for GIFANIM device.

GOPTIONS Components

```
goptions reset=all device=gifanim gsfname=plot1 gsfmode=replace delay=100  
cback=white cpattern=white ctext=black ftext=swissb ftitle=swissb  
xpixels=800 ypixels=600 iteration=0 gepilog='3B'X transparency  
disposal=background;
```

TRANSPARENCY = Specifies whether the background of the image should appear to be transparent when the image is displayed in the browser or PowerPoint.


GOPTIONS Components

```
goptions reset=all device=gifanim gsfname=plot1 gsfmode=replace delay=100  
cback=white cpattern=white ctext=black ftext=swissb ftitle=swissb  
xpixels=800 ypixels=600 iteration=0 gepilog='3B'X transparency  
disposal=background;
```

DISPOSAL = Specifies what happens to the graphic after it is displayed. 'background' causes the background color to be returned and the graph erased after displaying.

Closest 100

SU & 2001 Comparison Group Highlighted



Example – Rotating Plot

```
filename plot5 'file location & name.gif';

goptions reset=all device=gifanim gsfname=plot5 gsfmode=replace delay=2
        xpixels=1280 ypixels=800 iteration=0 gepilog='3B'X transparency
        disposal=background;

proc g3d data=su.top100;
    scatter gradrate4yr*pctftdoc=headcount / noneedle shape='balloon' size=.2 color=color6
        zmin=0 zmax=3500 zticknum=8 yticknum=5
        xticknum=4 grid rotate=20 to 80 by 1;

run;

quit;
```



Hans Rosling – TED.com Video

Virginia Public Four-Year Institutions

First-Time In-State Freshmen — Acceptance Rate vs. Enrollment Rate





Figure 30
 First-Time Freshmen Admissions
 In-State Acceptance and Enrollment Rates
 Fall 1973 through Fall 1987
 Four-Year Institutions






JMP Example

JMP Bubble Chart - Shockwave

Virginia Public Four-Year Institutions

First-Time In-State Freshmen — Acceptance Rate vs. Enrollment Rate



Example - Components

```
/* Develop additional data points for a smooth */
/* transition from actual point-to-point      */
data admis.admissions2;
  set admis.admissions1;
  by domicile inst year;
  lyear=lag(year);
  lenr=lag(erate);
  lacc=lag(arate);
  lapplied=lag(applied);
  if not first.inst then do;
    if erate=lenr then m=0;
    else m=(arate-lacc)/(erate-lenr);
    do i=0 to 1 by .04;
      yearx=lyear+i;
      aratex=lacc+i*(arate-lacc);
      eratex=lenr+i*(erate-lenr);
      appliedx=lapplied+i*(applied-lapplied);
      output;
    end;
  end;
run;
```


Example - Components

```
/* The following sort - 'descending appliedx' - */  
/* makes sure that smaller bubbles will always */  
/* be on top of larger bubbles */
```

```
proc sort data=admis.admissions2;  
  by domicile yearx descending appliedx;  
run;
```

```
/* Find minimum and maximum values to be used */  
/* in determining the relative bubble sizes */
```

```
proc means data=admis.admissions2 noprint;  
  var appliedx;  
  output out=admis.anno1x min=min1 max=max1;  
run;
```

Example - Components

```
/* Create institution bubbles */  
  
position='+';  
if adm and domicile='In-State' then do;  
  x=eratex;  
  y=aratex;  
  
/* Marker size is linear - bubble size is an area */  
  
  sizex=sizef*sqrt(appliedx/maxx);  
  function='symbol';  
  style='marker';  
  text='W';  
  size=sizex;
```

Example - Components

```
filename plot1 'file location & name.gif';

goptions reset=all device=gifanim gsfname=plot1 gsfmode=replace delay=10 cback=white
        cpattern=white xpixels=1280 ypixels=800 iteration=0 gepilog='3B'X transparency
        disposal=background;


axis1 order=(0 to 1 by .1) minor=none label=(font='swiss' angle=90 'Acceptance Rate')
        value=(font='swiss');

axis2 order=(0 to 1 by .1) minor=none label=(font='swiss' 'Enrollment Rate')
        value=(font='swiss');

options nobyline;

proc gplot data=admis.dummy;
  where domicile='In-State';
  by domicile yearx;
  plot aratex*eratex / annotate=admis.anno1a nolegend
        vaxis=axis1 vref=.5 lvref=2 vreverse
        haxis=axis2 href=.5 lhref=2 frame;
  title1 font='swiss' 'Virginia Public Four-Year Institutions';
  title2 font='swiss' 'First-Time In-State Freshmen - Acceptance Rate vs. Enrollment
Rate';
run;
quit;
```

Isoclines — Stable Enrollment





Example – Bubble (Browser)



Example – Storms (QuickTime)

Resources

- Robert Allison's SAS/Graph Examples! website (www.robslink.com/SAS/Home.htm)
- Mike Zdeb website (www.albany.edu/~msz03/)
- SAS Support (support.sas.com/)
 - SUGI Papers/Presentations
 - SAS Global Forum Papers/Presentations
 - SAS Papers
- TED.com
- GAPMINDER.org



Contact:

Jim Alessio

jim@alessios.net